
Miguel A. Gutiérrez Naranjo

Dpto. de Ciencias de la Computación e Inteligencia Artificial

Universidad de Sevilla

RA 2002–03 CcIa Implementación en Prolog de los tableros semánticos DA-8.1



Demostración por tableros semánticos

x Demostración de fórmula válida:
u Demostración:

¬p ∨ ¬q → ¬(p ∧ q) es válida

syss {¬(¬p ∨ ¬q → ¬(p ∧ q))} es inconsistente

syss {¬p ∨ ¬q, ¬¬(p ∧ q)} es inconsistente

syss {¬p ∨ ¬q, p ∧ q} es inconsistente

syss {p, q, ¬p ∨ ¬q} es inconsistente

syss {p, q, ¬p} es inconsistente y

{p, q, ¬q} es inconsistente

u Tablero semántico:

{−p v −q, p & q}

{−p v −q, −−(p & q)}

{−(−p v −q −> −(p & q))}

{−p v −q, p, q}

{−p, p, q} {−q, p, q}

Cerrado Cerrado

Demostración por tableros semánticos

x Demostración de fórmula no válida:
u Demostración:

¬p ∨ ¬q → ¬(p ∧ r) es válida

syss {¬(¬p ∨ ¬q → ¬(p ∧ r))} es inconsistente

syss {¬p ∨ ¬q, ¬¬(p ∧ r)} es inconsistente

syss {¬p ∨ ¬q, p ∧ r} es inconsistente

syss {p, q, ¬p ∨ ¬r} es inconsistente

syss {p, q, ¬p} es inconsistente y

{p, q, ¬r} es inconsistente

u Tablero semántico:

{−p v −q, p & r}

{−p v −q, −−(p & r)

{−(−p v −q −> −(p & r))}

{−p v −q, p, r}

{−p, p, r} {−q, p, r}

Cerrado Abierto

Notación uniforme

x Literales
u Un literal es un átomo o la negación de un átomo.

u literal(+F) se verifica si la fórmula F es un literal.

u Def. de literal:

literal(F) :- atom(F).
literal(-F) :- atom(F).

x Dobles negaciones

u La fórmula F es una doble negación si existe una

fórmula G tal que F es de la forma ¬¬G.

u Entonces |= F ↔ G.

x Fórmulas alfa
u Las fórmulas alfa, junto con sus componentes, son las

indicadas en la siguiente tabla
A1 ∧ A2 A1 A2

¬(A1 → A2) A1 ¬A2

¬(A1 ∨ A2) ¬A1 ¬A2

u Si F es una fórmula alfa y sus componentes son F1 y

F2, entonces |= F ↔ F1 ∧ F2.

u alfa(+A,-A1,-A2) se verifica si A es una fórmula alfa y

sus componentes son A1 y A2.

u Def. de alfa:

alfa(A1 & A2, A1, A2).
alfa(-(A1 => A2), A1, -A2).
alfa(-(A1 v A2), -A1, -A2).

Notación uniforme

x Fórmulas beta
u Las fórmulas beta, junto con sus componentes, son las

indicadas en la siguiente tabla
B1 ∨ B2 B1 B2

B1 → B2 ¬B1 B2

¬(B1 ∧ B2) ¬B1 ¬B2

B1 ↔ B2 B1 ∧ B2 ¬B1 ∧ ¬B2

¬(B1 ↔ B2) B1 ∧ ¬B2 ¬B1 ∧ B2

u Si F es una fórmula beta y sus componentes son F1 y

F2, entonces |= F ↔ F1 ∨ F2.

u beta(+B,-B1,-B2) se verifica si B es una fórmula beta y

sus componentes son B1 y B2.

u Def. de beta:

beta(B1 v B2, B1, B2).
beta(B1 => B2, -B1, B2).
beta(-(B1 & B2), -B1, -B2).
beta(B1 <=> B2, B1 & B2, -B1 & -B2).
beta(-(B1 <=> B2), B1 & -B2, -B1 & B2).

Completación de tableros

u Un tablero correspondiente a un conjunto de fórmulas

S es un árbol construido mediante las reglas:

(I) El árbol cuyo único nodo tiene como etiqueta S es

un tablero de S.

(C) Sea T un tablero de S y S1 la etiqueta de una hoja

de T .

(C.1) Si S1 cerrado (es decir, que contiene una

fórmula y su negación), entonces el árbol

obtenido añadiendo como hijo de S1 el nodo

etiquetado con cerrado es un tablero de S.

(C.2) Si S1 es abierto (es decir, es un conjunto de

literales que no contiene una fórmula y su ne-

gación), entonces el árbol obtenido añadiendo

como hijo de S1 el nodo etiquetado con abierto

es un tablero de S.

(C.3) Si S1 contiene una doble negación ¬¬F , en-

tonces el árbol obtenido añadiendo como hijo de

S1 el nodo etiquetado con (S1 − {¬¬F}) ∪ {F}
es un tablero de S.

(C.4) Si S1 contiene una fórmula alfa F de com-

ponentes F1 y F2, entonces el árbol obtenido

añadiendo como hijo de S1 el nodo etiquetado

con (S1 − {F}) ∪ {F1, F2} es un tablero de S.

(C.5) Si S1 contiene una fórmula beta F de com-

ponentes F1 y F2, entonces el árbol obtenido

añadiendo como hijos de S1 los nodos etiqueta-

dos con (S1 − {F}) ∪ {F1} y (S1 − {F}) ∪ {F2}
es un tablero de S.

Completación de tableros

u Un tablero semántico de S es completo si no se le

puede aplicar ninguna de las reglas de expansión; es

decir, todas sus hojas son abiertas o cerradas.

u tablero completo(+S,-Tab) se verifica si Tab es un

tablero completo del conjunto de fórmulas S.

u Ejemplo:

?- tablero_completo([- (-p v -q => - (p & r))],T).
T = t([- (-p v -q => - (p & r))],

t([-p v -q, - -(p & r)],
t([p & r, -p v -q],

t([p, r, -p v -q],
t([-p, p, r], cerrado, vacio),
t([-q, p, r], abierto, vacio)),

vacio),
vacio),

vacio)
Yes

u Representación: t(S,Izq,Dcha) representa el tablero

de ráız S, rama izquierda Izq y rama derecha Dcha.

u Def. de tablero completo:

tablero_completo(S,Tab) :-
completacion(t(S,_Izq,_Dcha),Tab).

Completación de tableros

u completacion(+Tab1,-Tab2) se verifica si Tab2 es una

completación (i.e. un tablero completo) del tablero

Tab1.

u Def. de completacion:

completacion(t(Flas,Izq1,Dcha1),t(Flas,Izq3,Dcha3)) :-
paso(t(Flas,Izq1,Dcha1),t(Flas,Izq2,Dcha2)), !,
completacion(Izq2,Izq3),
completacion(Dcha2,Dcha3).

completacion(Tab,Tab).

u paso(+Tab1,-Tab2) se verifica si Tab2 es un tablero

obtenido aplicando una regla de completación al

tablero Tab1.

u Def. de paso

paso(t(S1,_,_),t(S1, cerrado, vacio)) :- % C1
cerrada(S1), !.

paso(t(S1,_,_),t(S1, abierto, vacio)) :- % C2
lista_de_literales(S1), !.

paso(t(S1,_,_),t(S1, Izq, vacio)) :- % C3
regla_doble_negacion(S1, S2), !,
Izq = t(S2, _, _).

paso(t(S1,_,_),t(S1, Izq, vacio)) :- % C4
regla_alfa(S1, S2), !,
Izq = t(S2, _, _).

paso(t(S1,_,_),t(S1, Izq, Dcha)) :- % C5
regla_beta(S1, S2, S3),
Izq = t(S2, _, _),
Dcha = t(S3, _, _).

Completación de tableros

u cerrada(+S) se verifica si S es una lista cerrada (i.e.

que contiene una fórmula y su negación).

u Def. de cerrada:

cerrada(S) :- member(-F,S), member(F,S).

u lista de literales(+S) se verifica si S es una lista de

literales.

u Def. de lista de literales:

lista_de_literales([]).
lista_de_literales([F|S]) :-

literal(F),
lista_de_literales(S).

u regla doble negacion(+S1,-S2) que se verifica si S1 es

una lista de fórmulas que contiene una doble negación

--F y S2 es la lista de fórmulas obtenidas sustituyendo

en S1 la fórmula --F por F.

u Ejemplos:

?- regla_doble_negacion([- -(q v r), p => r],L).
L = [q v r, p => r]
?- regla_doble_negacion([p v (q v r), p => r],L).
No

u Def. de regla doble negacion:

regla_doble_negacion(S1, [F|S2]) :-
member(- -F, S1), !,
delete(S1, - -F, S2).

Completación de tableros
u regla alfa(+S1,-S2) que se verifica si S1 es una lista

de fórmulas que contiene una fórmula alfa A y S2 es

la lista de fórmulas obtenidas sustituyendo en S1 la

fórmula A por sus componentes.

u Ejemplos:

?- regla_alfa([p & (q v r), p => r],L).
L = [p, q v r, p => r]
?- regla_alfa([p v (q v r), p => r],L).
No

u Def. de regla alfa(+S1,-S2):

regla_alfa(S1, [A1,A2|S2]) :-
member(A, S1),
alfa(A, A1, A2), !,
delete(S1, A, S2).

u regla beta(+S1,-S2,-S3) que se verifica si S1 es una

lista de fórmulas que contiene al menos una fórmula

beta B y S2 es la lista de fórmulas obtenidas susti-

tuyendo en S1 la fórmula B por una de sus compo-

nentes y S3, por la otra.

u Ejemplos:

?- regla_beta([p & (q v r), p => r],L1,L2).
L1 = [-p, p & (q v r)]
L2 = [r, p & (q v r)]
?- regla_beta([p & (q v r), -(p => r)],L1,L2).
No

u Def. de regla beta:

regla_beta(S1, [B1|RS1], [B2|RS1]) :-
member(B, S1),
beta(B, B1, B2),
delete(S1, B, RS1).

Tableros cerrados

x Tableros cerrados
u Un tablero es cerrado si todas sus hojas están etique-

tadas con cerrado o vacio.

u es cerrado(+Tab) se verifica si Tab es un tablero cer-

rado.

u Def. de es cerrado:

es_cerrado(t(_,Izq,Dcha)) :-
es_cerrado(Izq),
es_cerrado(Dcha).

es_cerrado(cerrado).
es_cerrado(vacio).

Teorema por tableros

x Una fórmula F es un teorema (mediante
tableros semánticos) si tiene una prueba medi-
ante tableros; es decir, si {¬F} tiene un tablero
completo cerrado.

x Prueba mediante tableros
u prueba(+F,-Tab) se verifica si Tab es una prueba (me-

diante tableros semánticos) de la fórmula F.

u Ejemplos:

?- prueba(-p v -q => -(p & q),T).
T = t([- (-p v-q => - (p & q))],

t([-p v-q, - -(p & q)],
t([p & q, -p v -q],

t([p, q, -p v -q],
t([-p, p, q], cerrado, vacio),
t([-q, p, q], cerrado, vacio)),

vacio),
vacio),

vacio)
?- prueba(-p v -q => -p,T).
No

u Def. de prueba:

prueba(F,Tab) :-
tablero_completo([-F],Tab), !,
es_cerrado(Tab).

Teorema por tableros

x Teorema mediante tableros
u es teorema(+F) se verifica si la fórmula F es teorema

(mediante tableros semánticos).

u Ejemplos:

?- es_teorema(-p v -q => -(p & q)).
Yes
?- es_teorema(-p v -q => -(p & r)).
No

u Def. de es teorema:

es_teorema(F) :-
prueba(F,_Tab).

x El cálculo de tableros semánticos es adecuado
y completo; es decir, una fórmula es válida syss
es teorema mediante tableros semánticos.

Deducción por tableros

x La fórmula F es deducible (mediante tableros
semánticos) a partir del conjunto de fórmulas
S si existe una prueba mediante tableros de
que F a partir de S; es decir, existe un tablero
completo cerrado de S ∪ {¬F}.

x Deducción por tableros

u prueba deducible tab(+S,+F,-Tab) se verifica si Tab es

una prueba por tableros semánticos de que la fórmula

F es deducible del conjunto de fórmulas S.

u Ejemplos:

?- prueba_deducible_tab([p => q, q => r], p => r,T).
T = t([- (p => r), p => q, q => r],

t([p, -r, p => q, q => r],
t([-p, p, -r, q => r], cerrado, vacio),
t([q, p, -r, q => r],

t([-q, q, p, -r], cerrado, vacio),
t([r, q, p, -r], cerrado, vacio))),

vacio)
Yes
?- prueba_deducible_tab([p => q, q => r], p <=> r,T).
No

Deducción por tableros

Cerrado

Cerrado Cerrado

{−(p => r), p => q, q => r}

{p, −r, p => q, q => r}

{−p, p, −r, q => r} {q, p, −r, q => r}

{−q, q, p, −r} {r, q, p, −r}

u Def. de prueba deducible tab:

prueba_deducible_tab(S,F,Tab) :-
tablero_completo([-F|S],Tab), !,
es_cerrado(Tab).

x Deducibilidad por tableros

u es deducible tab(+S,+F) se verifica si la fórmula F

es deducible (mediante tableros) del conjunto de

fórmulas S.

u Def. de es deducible tab:

es_deducible_tab(S,F) :-
prueba_deducible_tab(S,F,_Tab).

x La fórmula F es consecuencia de S syss F es
deducible mediante tableros a partir de S.

